@echo off
call environment.bat passive

REM ============================
REM PYTHON PATH AUTO-DETECTION
REM for comfy or webui portable
REM ============================

set PY_EMBEDDED=%~dp0python_embeded
set PY_SYSTEM=%~dp0system\python

if exist "%PY_EMBEDDED%\python.exe" (
    pushd "%PY_EMBEDDED%"
    echo Detected python.exe in python_embedded
) else if exist "%PY_SYSTEM%\python.exe" (
    pushd "%PY_SYSTEM%"
    echo Detected python.exe in system\python
) else (
    echo ERROR: Could not find python.exe in:
    echo   %PY_EMBEDDED%
    echo   %PY_SYSTEM%
    echo.
    echo Please make sure one of these folders contains python.exe.
    pause
    exit /b 1
)

setlocal enabledelayedexpansion
:menu
cls
echo ========================================
echo   PyTorch Installation Selector
echo ========================================
echo.
echo This batch will remove the currently installed 
echo pytorch version and (re)install the current version
echo for the GPU type you are choosing.
echo.
echo WARNING! Only do this if you know what you are doing!
echo.
echo 1 ^) Install PyTorch for NVIDIA (CUDA 13.0)
echo 2 ^) Install PyTorch for NVIDIA (CUDA 12.8)
echo 3 ^) Install PyTorch for Intel (XPU)
REM echo 4 ^) Install PyTorch for AMD (ROCm) (experimental!)
echo X ^) Exit
echo.
set /p choice="Select an option (1-4): "

if "%choice%"=="1" goto install_nvidia130
if "%choice%"=="2" goto install_nvidia128
if "%choice%"=="3" goto install_intel
REM if "%choice%"=="4" goto install_amd
if "%choice%"=="X" exit /b

echo Invalid choice!
goto menu

:uninstall
echo.
echo Removing existing PyTorch installations...
.\python.exe -m pip uninstall -y torch torchvision torchaudio
echo ### Uninstall complete.
pause
goto :eof

:install_nvidia130
call :uninstall
echo ### Installing PyTorch for NVIDIA (CUDA 13.0) ...
.\python.exe -m pip install torch torchvision torchaudio --extra-index-url https://download.pytorch.org/whl/cu130
goto finish

:install_nvidia128
call :uninstall
echo ### Installing PyTorch for NVIDIA (CUDA 12.8) ...
.\python.exe -m pip install torch torchvision torchaudio --extra-index-url https://download.pytorch.org/whl/cu128
goto finish

:install_intel
call :uninstall
echo ### Installing PyTorch for Intel (XPU)
.\python.exe -m pip install torch torchvision torchaudio --extra-index-url https://download.pytorch.org/whl/xpu
goto finish

:install_amd
call :uninstall
echo ### Installing PyTorch for AMD...
.\python.exe -m pip install --no-cache-dir https://repo.radeon.com/rocm/windows/rocm-rel-7.1.1/rocm_sdk_core-0.1.dev0-py3-none-win_amd64.whl
.\python.exe -m pip install --no-cache-dir https://repo.radeon.com/rocm/windows/rocm-rel-7.1.1/rocm_sdk_devel-0.1.dev0-py3-none-win_amd64.whl
.\python.exe -m pip install --no-cache-dir https://repo.radeon.com/rocm/windows/rocm-rel-7.1.1/rocm_sdk_libraries_custom-0.1.dev0-py3-none-win_amd64.whl
.\python.exe -m pip install --no-cache-dir https://repo.radeon.com/rocm/windows/rocm-rel-7.1.1/rocm-0.1.dev0.tar.gz
.\python.exe -m pip install --no-cache-dir https://repo.radeon.com/rocm/windows/rocm-rel-7.1.1/torch-2.9.0+rocmsdk20251116-cp312-cp312-win_amd64.whl
.\python.exe -m pip install --no-cache-dir https://repo.radeon.com/rocm/windows/rocm-rel-7.1.1/torchaudio-2.9.0+rocmsdk20251116-cp312-cp312-win_amd64.whl
.\python.exe -m pip install --no-cache-dir https://repo.radeon.com/rocm/windows/rocm-rel-7.1.1/torchvision-0.24.0+rocmsdk20251116-cp312-cp312-win_amd64.whl
goto finish

:finish
echo Done!
pause